Chapter 26

Modeling Filters and Networks

Applying Kirchhoff's laws to circuits containing energy storage elements results in simultaneous differential equations in the time domain that must be solved to analyze the circuit's behavior. The solution of any equation of higher than first order can be difficult, and some driving functions cannot be solved easily by classical methods.

In both cases, the solution might be simplified using Laplace transforms to convert time domain equations containing integral and differential terms to algebraic equations in the frequency domain.

This chapter covers the following topics:

- Understanding Transient Modeling
- Using G and E Elements
- Modeling with Laplace and Pole-Zero
- Modeling Switched Capacitor Filters

Understanding Transient Modeling

The Laplace transform method also provides an easy way of relating a circuit's behavior in time and frequency-domains, facilitating simultaneous work in those domains.

The performance of the algorithm Star-Hspice uses for Laplace and pole/zero transient modeling is better than the performance of the Fast Fourier Transform (FFT) algorithm. Laplace and pole/zero transient modeling is invoked by using a LAPLACE or POLE function call in a source element statement.

Laplace transfer functions are especially useful in top-down system design, using ideal transfer functions instead of detailed circuit designs. Star-Hspice also allows you to mix Laplace transfer functions with transistors and passive components. Using this capability, a system may be modeled as the sum of the contributing ideal transfer functions, which can be progressively replaced by detailed circuit models as they become available. Laplace transfer functions are also conveniently used in control systems and behavioral models containing nonlinear elements.

Using Laplace transforms can reduce the long simulation times (as well as design time) of large interconnect systems, such as clock distribution networks, for which you can use methods such as asymptotic waveform evaluation (AWE) to create a Laplace transfer function model. The AWE model can represent the large circuit with just a few poles. You can input these poles through a Laplace transform model to closely approximate the delay and overshoot characteristics of many networks in a fraction of the original simulation time.

Pole/zero analysis is important in determining the stability of the design. The POLE function in Star-Hspice is useful when the poles and zeros of the circuit are provided, or they can be derived from the transfer function. (You can use the Star-Hspice .PZ statement to find poles and zeros. See "Using Pole/Zero Analysis" on page 24-3 for information about the .PZ statement).

Frequency response, an important analog circuit property, is normally specified as a ratio of two complex polynomials (functions of complex frequencies) with positive real coefficients. Frequency response can be given in the form of the locations of poles and zeros or can be in the form of a frequency table.

Complex circuits are usually designed by interconnecting smaller functional blocks of known frequency response, either in pole/zero or frequency table form. For example, you can design a band-reject filter by interconnecting a low-pass filter, a high-pass filter, and an adder. The designer should study the function of the complex circuit in terms of its component blocks before designing the actual circuit. After testing the functionality of the component blocks, they can be used as a reference in using optimization techniques to determine the complex element's value.

Using G and E Elements

This section describes how to use the G and E elements.

Laplace Transform Function Call

Use the Star-Hspice G and E elements (controlled behavioral sources) as linear functional blocks or elements with specific frequency responses in the following forms:

- Laplace transforms
- Pole/zero modeling
- Frequency response table

The frequency response is called the impulse response and is denoted by H(s), where s is a complex frequency variable ($s = j2\pi f$). In Star-Hspice, the frequency response is obtained by performing an AC analysis with AC=1 in the input source (the Laplace transform of an impulse is 1). The input and output of the G and E elements with specified frequency response are related by the expression:

$$Y(j2\pi f) = Hj(2\pi f) \cdot Xj(2\pi f)$$

where X, Y and H are the input, the output, and the transfer function at frequency f.

For AC analysis, the frequency response is determined by the above relation at any frequency. For operating point and DC sweep analysis, the relation is the same, but the frequency is zero.

The transient analysis is more complicated than the frequency response. The output is a convolution of the input waveform with the impulse response h(t):

$$y(t) = \int_{-\infty}^{\infty} x(\tau) \cdot h(t-\tau) \cdot d\tau$$

In discrete form, the output is

$$y(k\Delta) = \Delta \sum_{m=0}^{k} x(m\Delta) \cdot h[(k-m) \cdot \Delta] \ , \ k = 0, 1, 2, ...$$

where the h(t) can be obtained from H(f) by the inverse Fourier integral:

$$h(t) = \int_{-\infty}^{\infty} H(f) \cdot e^{j2\pi ft} \cdot df$$

The inverse discrete Fourier transform is given by

$$h(m\Delta) = \frac{1}{N \cdot \Delta} \sum_{n=0}^{N-1} H(f_n) \cdot e^{\frac{j2\pi nm}{N}} \ , \ m = 0, 1, 2, ..., N\text{-}1$$

where N is the number of equally spaced time points and Δ is the time interval or time resolution.

For the frequency response table form (FREQ) of the LAPLACE function, Star-Hspice‘s performance-enhanced algorithm is used to convert H(f) to h(t). This algorithm requires N to be a power of 2. The frequency point f_n is determined by

$$f_n = \frac{n}{N \cdot \Delta} \ , \quad n = 0, 1, 2, ..., N\text{-}1$$

where n > N/2 represents the negative frequencies. The Nyquist critical frequency is given by

$$f_c = f_{N/2} = \frac{1}{2 \cdot \Delta}$$

Since the negative frequencies responses are the image of the positive ones, only N/2 frequency points are required to evaluate N time points of h(t). The larger f_c is, the more accurate the transient analysis results are. However, for large f_c, the Δ becomes smaller, and computation time increases. The maximum frequency of interest depends on the functionality of the linear network. For example, in a low-pass filter, f_c can be set to the frequency at which the response drops by 60 dB (a factor of 1000).

$$|H(f_c)| = \frac{|H_{max}|}{1000}$$

Once f_c is selected or calculated, then Δ can be determined by

$$\Delta = \frac{1}{2 \cdot f_c}$$

Notice the frequency resolution

$$\Delta f = f_1 = \frac{1}{N \cdot \Delta}$$

is inversely proportional to the maximum time (N·Δ) over which h(t) is evaluated. Therefore, the transient analysis accuracy also depends on the frequency resolution or the number of points (N).You can specify the frequency resolution DELF and maximum frequency MAXF in the G or E element statement. N is calculated by 2·MAXF/DELF. Then, N is modified to be a power of 2. The effective DELF is determined by 2·MAXF/N to reflect the changes in N.

Laplace Transform – LAPLACE Function

The syntax is:

Transconductance H(s):

Gxxx n_+ n_- LAPLACE in_+ in_- k_0, k_1, ..., k_n / d_0, d_1, ..., d_m

+ <SCALE=val> <TC1=val> <TC2=val> <M=val>

Voltage Gain H(s):

Exxx n_+ n_- LAPLACE in_+ in_- k_0, k_1, ..., k_n / d_0, d_1, ..., d_m

<SCALE=val> <TC1=val> <TC2=val>

H(s) is a rational function in the following form:

$$H(s) = \frac{k_0 + k_1 s + \ldots + k_n s^n}{d_0 + d_1 s + \ldots + d_m s^m}$$

All the coefficients k_0, k_1, ..., d_0, d_1, ..., can be parameterized.

Examples

```
Glowpass  0  out  LAPLACE  in  0   1.0 / 1.0  2.0  2.0  1.0
Ehipass    out  0    LAPLACE  in  0   0.0,0.0,0.0,1.0 /
1.0,2.0,2.0,1.0
```

The Glowpass element statement describes a third-order low-pass filter with the transfer function

$$H(s) = \frac{1}{1 + 2s + 2s^2 + s^3}$$

The Ehipass element statement describes a third-order high-pass filter with the transfer function

$$H(s) = \frac{s^3}{1 + 2s + 2s^2 + s^3}$$

Laplace Transform – Pole-Zero Function

General Forms

Transconductance H(s):

Gxxx n_+ n_- POLE in_+ in_- a α_{z1}, f_{z1}, ..., α_{zn}, f_{zn} / b, α_{p1}, f_{p1}, ..., α_{pm}, f_{pm}

+ <SCALE=val> <TC1=val> <TC2=val> <M=val>

Voltage Gain H(s):

Exxx n_+ n_- POLE in_+ in_- a α_{z1}, f_{z1}, ..., α_{zn}, f_{zn} / b, α_{p1}, f_{p1}, ..., α_{pm}, f_{pm}

+ <SCALE=val> <TC1=val> <TC2=val>

H(s) in terms of poles and zeros is defined by

$$H(s) = \frac{a \cdot (s + \alpha_{z1} - j2\pi f_{z1}) \ldots (s + \alpha_{zn} - j2\pi f_{zn})(s + \alpha_{zn} + j2\pi f_{zn})}{b \cdot (s + \alpha_{p1} - j2\pi f_{p1}) \ldots (s + \alpha_{pm} - j2\pi f_{pm})(s + \alpha_{pm} + j2\pi f_{pm})}$$

Notice the complex poles or zeros are in conjugate pairs. In the element description, only one of them is specified, and the program includes the conjugate. The a, b, α, and f values can be parameterized.

Examples

```
Ghigh_pass 0 out POLE in 0 1.0 0.0,0.0 / 1.0 0.001,0.0
Elow_pass out 0 POLE in 0 1.0 / 1.0, 1.0,0.0 0.5,0.1379
```

The Ghigh_pass statement describes a high-pass filter with transfer function

$$H(s) = \frac{1.0 \cdot (s + 0.0 + j \cdot 0.0)}{1.0 \cdot (s + 0.001 + j \cdot 0.0)}$$

The Elow_pass statement describes a low-pass filter with transfer function

$$H(s) = \frac{1.0}{1.0 \cdot (s + 1)(s + 0.5 + j2\pi \cdot 0.1379)(s + 0.5 - (j2\pi \cdot 0.1379))}$$

Laplace Transform- Frequency Response Table

The syntax is:

Transconductance H(s):

Gxxx n+ n- FREQ in+ in- $f_1, a_1, \phi_1, \ldots, f_i, a_i, \phi_1$

+ <DELF=val> <MAXF=val> <SCALE=val> <TC1=val>

+ <TC2=val> <M=val> <LEVEL=val>

Voltage Gain H(s):

Exxx n+ n- FREQ in+ in- $f_1, a_1, \phi_1, \ldots, f_i, a_i, \phi_1$

+ <DELF=val> <MAXF=val> <SCALE=val> <TC1=val> <TC2=val>

Each f_i is a frequency point in hertz, a_i is the magnitude in dB, and ϕ_1 is the phase in degrees. At each frequency the network response, magnitude, and phase are calculated by interpolation. The magnitude (in dB) is interpolated logarithmically as a function of frequency. The phase (in degrees) is interpolated linearly as a function of frequency.

$$|H(j2\pi f)| = \left(\frac{a_i - a_k}{\log f_i - \log f_k}\right)(\log f - \log f_i) + a_i$$

$$\angle H(j2\pi f) = \left(\frac{\phi_i - \phi_k}{f_i - f_k}\right)(f - f_i) + \phi_i$$

Example

```
Eftable  output   0  FREQ  input   0
+  1.0k   -3.97m   293.7
+  2.0k   -2.00m   211.0
+  3.0k   17.80m   82.45
+  .........
+ 10.0k  -53.20    -1125.5
```

The first column is frequency in hertz, the second is magnitude in dB, and third is phase in degrees. The LEVEL must be set to 1 for a high-pass filter, and the last frequency point must be the highest frequency response value that is a real number with zero phase. The frequency, magnitude, and phase in the table can be parameterized.

Element Statement Parameters

These keywords are common to the three forms, Laplace, pole-zero, and frequency response table described above.

DELF, DELTA frequency resolution Δf. The inverse of DELF is the time window over which h(t) is calculated from H(s). The smaller DELF is, the more accurate the transient analysis, and the longer the CPU time. The number of points, N, used in the conversion of H(s) to h(t) is N=2·MAXF/DELF. Since N must be a power of 2, the DELF is adjusted. The default is 1/TSTOP.

FREQ keyword to indicate that the transfer function is described by a frequency response table. Do not use FREQ as a node name in a G or E element.

LAPLACE keyword to indicate the transfer function is described by a Laplace transform function. Do not use LAPLACE as a node name on a G or E element.

LEVEL used only in elements with frequency response table. This parameter must be set to 1 if the element represents a high-pass filter.

M G element multiplier. This parameter is used to represent *M* G elements in parallel. Default is 1.

MAXF, MAX maximum or the Nyquist critical frequency. The larger the MAXF the more accurate the transient results and the longer is the CPU time. The default is $1024 \cdot DELF$. Thee parameters are applicable only when the FREQ parameter is also used.

POLE keyword to indicate the transfer function is described by the pole and zero location. Do not use POLE as a node name on a G or E element.

SCALE element value multiplier

TC1,TC2 first and second order temperature coefficients. The default is zero. The SCALE is updated by temperature:

$$ALEeff = SCALE \cdot (1 + TC1 \cdot \Delta t + TC2 \cdot \Delta$$

Note: Pole/zero analysis is not allowed when the data file contains elements with frequency response specifications
If you include a MAXF=<par> specification in a G or Element statement, Star-Hspice issues a warning that MAXF is ignored. This is normal.

Laplace Band-Reject Filter

This example models an active band-reject filter[1] with 3-dB points at 100 and 400 Hz and greater than 35 dB of attenuation between 175 and 225 Hz. The band-reject filter is made up of low-pass and high-pass filters and an adder. The low-pass and high-pass filters are fifth order Chebyshev with a 0.5-dB ripple.



Figure 26-1: Band-Reject Filter

Example Band-Reject Filter

```
BandstopL.sp band_reject filter
.OPTIONS PROBE POST=2
.AC DEC 50 10 5k
.PROBE AC VM(out_low) VM(out_high) VM(out)
.PROBE AC VP(out_low) VP(out_high) VP(out)
.TRAN .01m 12m
.PROBE V(out_low) V(out_high) V(out)
.GRAPH v(in)  V(out)
Vin in 0 AC 1 SIN(0,1,250)
```

Band_Reject Filter Circuit

```
Elp3 out_low3  0  LAPLACE  in 0
+ 1 / 1 6.729m 15.62988u 27.7976n
Elp  out_low   0  LAPLACE  out_low3 0
+ 1 / 1 0.364m 2.7482u
Ehp3 out_high3 0  LAPLACE  in 0
+ 0,0,0,9.261282467p /
+ 1,356.608u,98.33419352n,9.261282467p
Eph  out_high  0  LAPLACE  out_high3 0
```

```
+ 0 0 144.03675n / 1 83.58u 144.03675n
Eadd   out 0  VOL='-V(out_low) - V(out_high)'
Rl out 0 1e6
.END
```



Figure 26-2: Frequency Response of the Band-Reject Filter



Figure 26-3: Transient Response of the Band-Reject Filter to a 250 Hz Sine Wave

Laplace Low-Pass Filter

This example simulates a third-order low-pass filter with a Butterworth transfer function, comparing the results of the actual circuit and the functional G element with third-order Butterworth transfer function for AC and transient analysis.



Figure 26-4: Third-Order Active Low-Pass Filter

The third-order Butterworth transfer function that describes the above circuit is:

$$H(s) = \frac{1.0}{1.0 \cdot (s+1)(s+0.5+j2\pi \cdot 0.1379)(s+0.5-(j2\pi \cdot 0.1379))}$$

The following is the input listing of the above filter. Notice the pole locations are parameterized on the G element. Also, only one of the complex poles is specified. The conjugate pole is derived by the program. The output of the circuit is node “out” and the output of the functional element is “outg”.

Example Third-Order Low-Pass Butterworth Filter

```
Low_Pass.sp 3rd order low-pass Butterworth
.OPTIONS  POST=2 PROBE INTERP=1 DCSTEP=1e8
.PARAM a=1.0 b=1.0 ap1=1.0 fp1=0.0 ap2=0.5 fp2=0.1379
.AC DEC 25 0.01 10
.PROBE AC VDB(out) VDB(outg) VP(out) VP(outg)
.TRAN .5 200
.PROBE V(in) V(outg) V(out)
.GRAPH  V(outg) V(out)
VIN in 0  AC 1   PULSE(0,1,0,1,1,48,100)
* 3rd order low-pass described by G element
Glow_pass 0 outg POLE in 0 a / b ap1,fp1 ap2,fp2
Rg outg 0 1
```

Circuit Description

```
R1 in 2 1
R2 2  3 1
R3 3  4 1
C1 2  0 1.392
C2 4  0 0.2024
C3 3  out 3.546
Eopamp out 0   OPAMP  4 out
.END
```



Figure 26-5: Frequency Response of Circuit and Functional Element



Figure 26-6: Transient Response of Circuit and Functional Element to a Pulse

Modeling with Laplace and Pole-Zero

The Laplace Transform (LAPLACE) Function

There are two forms of the Star-Hspice LAPLACE function call, one for transconductance and one for voltage gain transfer functions. See “Using G and E Elements” on page -4 for the general forms and descriptions of the parameters.

General Form of the Transfer Function

To use the Star-Hspice LAPLACE modeling function, you must find the k_0, ..., k_n and d_0, ..., d_m coefficients of the transfer function. The transfer function is the s-domain (frequency domain) ratio of the output of a single-source circuit to the input, with initial conditions set to zero. The Laplace transfer function is represented by

$$H(s) = \frac{Y(s)}{X(s)},$$

where *s* is the complex frequency $j2\pi f$, *Y(s)* is the Laplace transform of the output signal, and *X(s)* is the Laplace transform of the input signal.

Note: In Star-Hspice, the impulse response H(s) is obtained by performing an AC analysis, with AC=1 representing the input source. The Laplace transform of an impulse is 1
For an element with an infinite response at DC, such as a unit step function H(s)=1/s, Star-Hspice uses the value of the EPSMIN option (the smallest number possible on the platform) for the transfer function in its calculations.

The general form of the transfer function *H(s)* in the frequency domain is

$$H(s) = \frac{k_0 + k_1 s + \ldots + k_n s^n}{d_0 + d_1 s + \ldots + d_m s^m}$$

The order of the numerator of the transfer function cannot be greater than the order of the denominator, except for differentiators, for which the transfer function $H(s) = ks$. All of the transfer function's k and d coefficients can be parameterized in the Star-Hspice circuit descriptions.

Finding the Transfer Function

The first step in determining the transfer function of a circuit is to convert the circuit to the *s*-domain by transforming each element's value into its *s*-domain equivalent form.

Tables 26-1 and 26-2 show transforms used to convert some common functions to the *s*-domain[2,3]. The next section provides examples of using transforms to determine transfer functions.

Table 26-1: Laplace Transforms for Common Source Functions

f(t), t>0	Source Type	L { f(t) }= F(s)
$\delta(t)$	impulse	1
$u(t)$	step	$\frac{1}{s}$
t	ramp	$\frac{1}{s^2}$
e^{-at}	exponential	$\frac{1}{s+a}$
$\sin \omega t$	sine	$\frac{\omega}{s^2+\omega^2}$
$\cos \omega t$	cosine	$\frac{s}{s^2+\omega^2}$

Table 26-1: Laplace Transforms for Common Source Functions

f(t), t>0	Source Type	L { f(t) }= F(s)
$\sin(\omega t+\theta)$	sine	$\frac{s\sin(\theta)+\omega\cos(\theta)}{s^2+\omega^2}$
$\cos(\omega t+\theta)$	cosine	$\frac{s\cos(\theta)-\omega\sin(\theta)}{s^2+\omega^2}$
$\sinh \omega t$	hyperbolic sine	$\frac{\omega}{s^2-\omega^2}$
$\cosh \omega t$	hyperbolic cosine	$\frac{s}{s^2-\omega^2}$
te^{-at}	damped ramp	$\frac{1}{(s+a)^2}$
$e^{-at}\sin \omega t$	damped sine	$\frac{\omega}{(s+a)^2+\omega^2}$
$e^{-at}\cos \omega t$	damped cosine	$\frac{s+a}{(s+a)^2+\omega^2}$

Table 26-2: Laplace Transforms for Common Operations

f(t)	L { f(t) } = F(s)
$Kf(t)$	$KF(s)$
$f_1(t)+f_2(t)-f_3(t)+\ldots$	$F_1(s)+F_2(s)-F_3(s)+\ldots$
$\frac{d}{dt}f(t)$	$sF(s)-f(0-\ )$

Table 26-2: Laplace Transforms for Common Operations

f(t)	L { f(t) } = F(s)
$\frac{d^2}{dt^2} f(t)$	$s^2 F(s) - sf(0) - \frac{d}{dt} f(0- \quad)$
$\frac{d^n}{dt^n} f(t)$	$s^n F(s) - s^{n-1} f(0) - s^{n-2} \frac{d}{dt} f(0)$
$\int_{-\infty}^{t} f(t) dt$	$\frac{F(s)}{s} + \frac{f^{-1}(0)}{s}$
$f(t-a)u(t-a), a > 0$ (u is the step function)	$e^{-as} F(s)$
$e^{-at} f(t)$	$F(s+a)$
$f(at), a > 0$	$\frac{1}{a} F\left(\frac{s}{a}\right)$
$tf(t)$	$-\frac{d}{ds}(F(s))$
$t^n f(t)$	$-(1)^n \frac{d^n}{ds^n} F(s)$
$\frac{f(t)}{t}$	$\int_{s}^{\infty} F(u) du$ (u is the step function)
$f(t-t_1)$	$e^{-t_1 s} F(s)$

Determining the Laplace Coefficients

The following examples describe how to determine the appropriate coefficients for the Laplace modeling function call in Star-Hspice.

LAPLACE Example 1 – Voltage Gain Transfer Function

To find the voltage gain transfer function for the circuit in Figure 26-7:, convert the circuit to its equivalent *s*-domain circuit and solve for v_o / v_g.



Figure 26-7: LAPLACE Example 1 Circuit

Use transforms from Table 26-2: to convert the inductor, capacitor, and resistors. $L\{f(t)\}$ represents the Laplace transform of $f(t)$:

$$L\left\{L\frac{d}{dt}f(t)\right\} = L \cdot (sF(s) - f(0)) = 50 \times 10^{-3} \cdot (s - 0) = 0.05s$$

$$L\left\{\frac{1}{C}\int_0^t f(t)d\tau\right\} = \frac{1}{C} \cdot \left(\frac{F(s)}{s} + \frac{f^{-1}(0)}{s}\right) = \frac{1}{10^{-6}} \cdot \left(\frac{1}{s} + 0\right) = \frac{10^6}{s}$$

$$L\{R1 \cdot f(t)\} = R1 \cdot F(s) = R1 = 1000 \quad \Omega$$

$$L\{R2 \cdot f(t)\} = R2 \cdot F(s) = R2 = 250 \quad \Omega$$

To convert the voltage source to the *s*-domain, use the sin ωt transform from Table 26-1:

$$L\{2\sin 3t\} = 2 \cdot \frac{3}{s^2 + 3^2} = \frac{6}{s^2 + 9}$$

Figure 26-8: displays the s-domain equivalent circuit.



Figure 26-8: S-Domain Equivalent of the LAPLACE Example 1 Circuit

Summing the currents leaving node n2:

$$\frac{v_o - v_g}{1000} + \frac{v_o}{250 + 0.05s} + \frac{v_o s}{10^6} = 0$$

Solve for v_o:

$$v_o = \frac{1000(s + 5000)v_g}{s^2 + 6000s + 25 \times 10^6}$$

The voltage gain transfer function is

$$H(s) = \frac{v_o}{v_g} = \frac{1000(s + 5000)}{s^2 + 6000s + 25 \times 10^6} = \frac{5 \times 10^6 + 1000s}{25 \times 10^6 + 6000s + s^2}$$

For the Star-Hspice Laplace function call, use k_n and d_m coefficients for the transfer function in the form:

$$H(s) = \frac{k_0 + k_1 s + \ldots + k_n s^n}{d_0 + d_1 s + \ldots + d_m s^m}$$

The coefficients from the voltage gain transfer function above are

$k_0 = 5 \times 10^6$ $\quad k_1 = 1000$

$d_0 = 25 \times 10^6$ $\quad d_1 = 6000$ $\quad d_2 = 1$

Using these coefficients, a Star-Hspice Laplace modeling function call for the voltage gain transfer function of the circuit in Figure 26-7: is

```
Eexample1 n1 n0 LAPLACE n2 n0 5E6 1000 / 25E6 6000 1
```

LAPLACE Example 2 – Differentiator

You can model a differentiator using either G or E elements as shown in the following example.

In the frequency domain:

E element: $V_{\text{out}} = ksV_{\text{in}}$

G element: $I_{\text{out}} = ksV_{\text{in}}$

In the time domain:

E element: $v_{\text{out}} = k\frac{dV_{\text{in}}}{dt}$

G element: $i_{\text{out}} = k\frac{dV_{\text{in}}}{dt}$

For a differentiator, the voltage gain transfer function is

$$H(s) = \frac{V_{out}}{V_{in}} = ks$$

In the general form of the transfer function,

$$H(s) = \frac{k_0 + k_1 s + \ldots + k_n s^n}{d_0 + d_1 s + \ldots + d_m s^m},$$

If you set $k_1 = k$ and $d_0 = 1$ and the remaining coefficients are zero, then the equation becomes

$$H(s) = \frac{ks}{1} = ks$$

Using the coefficients $k_1 = k$ and $d_0 = 1$ in the Laplace modeling, the Star-Hspice circuit descriptions for the differentiator are:

```
Edif out GND LAPLACE in GND 0 k / 1
Gdif out GND LAPLACE in GND 0 k / 1
```

LAPLACE Example 3 – Integrator

An integrator can be modeled by G or E elements as follows:

In the frequency domain:

$$\textit{E Element:}\quad V_{out} = \frac{k}{s} V_{in}$$

$$\textit{G Element:}\quad I_{out} = \frac{k}{s} V_{in}$$

In the time domain:

$$E\ Element: \; v_{out} = k \int V_{in} dt$$

$$G\ Element: \; i_{out} = k \int V_{in} dt$$

For an integrator, the voltage gain transfer function is:

$$H(s) = \frac{V_{out}}{V_{in}} = \frac{k}{s}$$

In the general form of the transfer function,

$$H(s) = \frac{k_0 + k_1 s + \dots + k_n s^n}{d_0 + d_1 s + \dots + d_m s^m}$$

Like the previous example, if you make $k_0 = k$ and $d_1 = 1$, then the equation becomes

$$H(s) = \frac{k + 0 + \dots + 0}{0 + s + \dots + 0} = \frac{k}{s}$$

Laplace Transform POLE (Pole/Zero) Function

This section describes the general form of the pole/zero transfer function and provides examples of converting specific transfer functions into pole/zero circuit descriptions.

The POLE Function Call

The POLE function in Star-Hspice is useful when the poles and zeros of the circuit are available. The poles and zeros can be derived from the transfer function, as described in this chapter, or you can use the Star-Hspice.PZ statement to find them, as described in “Using Pole/Zero Analysis” on page 24-3.

There are two forms of the Star-Hspice LAPLACE function call, one for transconductance and one for voltage gain transfer functions. See “Using G and E Elements” on page -4 for the general forms and list of optional parameters.

To use the POLE pole/zero modeling function, find the *a*, *b*, *f*, and α coefficients of the transfer function. The transfer function is the *s*-domain (frequency domain) ratio of the output of a single-source circuit to the input, with initial conditions set to zero.

General Form of the Transfer Function

The general expanded form of the pole/zero transfer function *H(s)* is:

$$H(s) = \frac{a(s + \alpha_{z1} + j2\pi f_{z1})(s + \alpha_{z1} - j2\pi f_{z1})\ldots(s + \alpha_{zn} + j2\pi f_{zn})(s + \alpha_{zn} - j2\pi f_{zn})}{b(s + \alpha_{p1} + j2\pi f_{p1})(s + \alpha_{p1} - j2\pi f_{p1})\ldots(s + \alpha_{pm} + j2\pi f_{pm})(s + \alpha_{pm} - j2\pi f_{pm})} \tag{1}$$

The a, b, α, and f values can be parameterized.

Examples

```
Ghigh_pass  0  out   POLE  in  0  1.0   0.0,0.0 / 1.0
0.001,0.0
Elow_pass  out  0     POLE  in  0   1.0 / 1.0,  1.0,0.0
0.5,0.1379
```

The Ghigh_pass statement describes a high pass filter with transfer function

$$H(s) = \frac{1.0 \cdot (s + 0.0 + j \cdot 0.0)}{1.0 \cdot (s + 0.001 + j \cdot 0.0)}$$

The Elow_pass statement describes a low-pass filter with transfer function

$$H(s) = \frac{1.0}{1.0 \cdot (s + 1)(s + 0.5 + j2\pi \cdot 0.1379)(s + 0.5 - (j2\pi \cdot 0.1379))}$$

To write an Star-Hspice pole/zero circuit description for an element, you need to know the element's transfer function *H(s)* in terms of the *a*, *b*, *f*, and α coefficients. Use the values of these coefficients in POLE function calls in the Star-Hspice circuit description.

First, however, simplify the transfer function, as described in the next section.

Star-Hspice Reduced Form of the Transfer Function

Complex poles and zeros occur in conjugate pairs (a set of complex numbers differ only in the signs of their imaginary parts):

$$(s + \alpha_{pm} + j2\pi f_{pm})(s + \alpha_{pm} - j2\pi f_{pm}) \text{, for poles}$$

and

$$(s + \alpha_{zn} + j2\pi f_{zn})(s + \alpha_{zn} - j2\pi f_{zn}) \text{, for zeros}$$

To write the transfer function in Star-Hspice pole/zero format, supply coefficients for one term of each conjugate pair and Star-Hspice provides the coefficients for the other term. If you omit the negative complex roots, the result is the reduced form of the transfer function, *Reduced*{*H*(*s*)}. Find the reduced form by collecting all the general form terms with negative complex roots:

$$H(s) = \frac{a(s + \alpha_{z1} + j2\pi f_{z1})\ldots(s + \alpha_{zn} + j2\pi f_{zn})}{b(s + \alpha_{p1} + j2\pi f_{p1})\ldots(s + \alpha_{pm} + j2\pi f_{pm})} \cdot \frac{a(s + \alpha_{z1} - j2\pi f_{z1})\ldots(s + \alpha_{zn} -}{b(s + \alpha_{p1} - j2\pi f_{p1})\ldots(s + \alpha_{pm} -} \quad (1)$$

Then discard the right-hand term, which contains all the terms with negative roots. What remains is the reduced form:

$$Reduced\{H(s)\} = \frac{a(s + \alpha_{z1} + j2\pi f_{z1})\ldots(s + \alpha_{zn} + j2\pi f_{zn})}{b(s + \alpha_{p1} + j2\pi f_{p1})\ldots(s + \alpha_{pm} + j2\pi f_{pm})} \quad (2)$$

For this function find the *a*, *b*, *f*, and α coefficients to use in an Star-Hspice POLE function for a voltage gain transfer function. The following examples show how to determine the coefficients and write POLE function calls for a high-pass filter and a low-pass filter.

POLE Example 1 – Highpass Filter

For a high-pass filter with a given transconductance transfer function, such as

$$H(s) = \frac{s}{(s + 0.001)}$$

Find the *a*, *b*, α, and *f* coefficients necessary to write the transfer function in the general form (1) shown previously, so that you can clearly see the conjugate pairs of complex roots. You only need to supply one of each conjugate pair of roots in the Laplace function call. Star-Hspice automatically inserts the other root.

To get the function into a form more similar to the general form of the transfer function, rewrite the given transconductance transfer function as

$$H(s) = \frac{1.0(s + 0.0)}{1.0(s + 0.001)}$$

Since this function has no negative imaginary parts, it is already in the Star-Hspice reduced form (2) shown previously. Now you can identify the *a*, *b*, *f*, and α coefficients so that the transfer function *H*(*s*) matches the reduced form. This matching process obtains the following values:

$n = 1, m = 1,$

$a = 1.0$ $\alpha_{z1} = 0.0$ $f_{z1} = 0.0$

$b = 1.0$ $\alpha_{p1} = 0.001$ $f_{p1} = 0.0$

Using these coefficients in the reduced form provides the desired transfer function, $\frac{s}{(s + 0.001)}$.

So the general transconductance transfer function POLE function call,

```
Gxxx n+ n- POLE in+ in- a αz1,fz1...αzn,fzn / b αp1,fp1...αpm,fpm
```

for an element named *Ghigh_pass* becomes

```
Ghigh_pass gnd out POLE in gnd 1.0 0.0,0.0 / 1.0 0.001,0.0
```

POLE Example 2 – Low-Pass Filter

For a low-pass filter with the given voltage gain transfer function

$$H(s) = \frac{1.0}{1.0(s + 1.0 + j2\pi \cdot 0.0)(s + 0.5 + j2\pi \cdot 0.15)(s + 0.5 - j2\pi \cdot 0.15)}$$

you need to find the *a*, *b*, α, and *f* coefficients to write the transfer function in the general form, so that you can identify the complex roots with negative imaginary parts.

To separate the reduced form, *Reduced*{*H*(*s*)}, from the terms with negative imaginary parts, rewrite the given voltage gain transfer function as

$$H(s) = \frac{1.0}{1.0(s + 1.0 + j2\pi \cdot 0.0)(s + 0.5 + j2\pi \cdot 0.15)} \cdot \frac{1.0}{(s + 0.5 - j2\pi \cdot 0.15)}$$

$$= Reduced\{H(s)\} \cdot \frac{1.0}{(s + 0.5 - j2\pi \cdot 0.15)}$$

So

$$Reduced\{H(s)\} = \frac{1.0}{1.0(s + 1.0)(s + 0.5 + j2\pi \cdot 0.15)}$$

or

$$\frac{a(s + \alpha_{z1} + j2\pi f_{z1})\ldots(s + \alpha_{zn} + j2\pi f_{zn})}{b(s + \alpha_{p1} + j2\pi f_{p1})\ldots(s + \alpha_{pm} + j2\pi f_{pm})} = \frac{1.0}{1.0(s + 1.0 + j2\pi \cdot 0.0)(s + 0.5 + j2\pi}$$

Now assign coefficients in the reduced form to match the given voltage transfer function. The following coefficient values produce the desired transfer function:

n = 0, m = 2,

a =1.0 *b = 1.0* $\alpha_{p1} = 1.0$ $f_{p1} = 0$ $\alpha_{p2} = 0.5 f_{p2} = 0.15$

These coefficients can be substituted in the POLE function call for a voltage gain transfer function,

```
Exxx n+ n- POLE in+ in- a αz1,fz1...αzn,fzn / b αp1,fp1...αpm,fpm
```

for an element named *Elow_pass* to obtain the Star-Hspice statement

```
Elow_pass out GND POLE in 1.0 / 1.0 1.0,0.0 0.5,0.15
```

RC Line Modeling

Most RC lines can have very simple models, with just a single dominant pole. The dominant pole can be found by AWE methods, computed based on the total series resistance and capacitance[4], or determined by the Elmore delay[5].

The Elmore delay uses the value (d1-k1) as the time constant of a single-pole approximation to the complete H(*s*), where H(*s*) is the transfer function of the RC network to a given output. The inverse Laplace transform of h(*t*) is H(*s*):

$$\tau_{DE} = \int_0^{\infty} t \cdot h(t)dt$$

Actually, the Elmore delay is the first moment of the impulse response, and so corresponds to a first order AWE result.



Figure 26-9: Circuits for an RC Line

RC Line Circuit File

```
* Laplace testing RC line
.Tran 0.02ns 3ns
.Options Post Accurate List Probe
v1 1 0 PWL 0ns 0 0.1ns 0 0.3ns 5 1.3ns 5 1.5ns 0
r1 1 2 200
c1 2 0 0.6pF
r2 2 3 80
c2 3 0 0.8pF
r3 3 4 160
c3 4 0 0.7pF
r4 4 5 200
c4 5 0 0.8pF
e1 6 0 LAPLACE 1 0 1 / 1 1.16n
.Probe v(1) v(5) v(6)
.Print v(1) v(5) v(6)
.End
```

The output of the RC circuit shown in Figure 26-9: can be closely approximated by a single pole response, as shown in Figure 26-10:.



Figure 26-10: Transient Response of the RC Line and Single-Pole Approximation

Notice in Figure 26-10: that the single pole approximation has less delay: 1 ns compared to 1.1 ns for the full RC line model at 2.5 volts. The single pole approximation also has a lower peak value than the RC line model. All other things being equal, a circuit with a shorter time constant results in less filtering and allows a higher maximum voltage value. The single-pole approximation produces a lower amplitude and less delay than the RC line because the single pole neglects the other three poles in the actual circuit. However, a single-pole approximation still gives very good results for many problems.

AWE Transfer Function Modeling

Single-pole transfer function approximations can cause larger errors for low-loss lines than for RC lines since lower resistance allows ringing. Because circuit ringing creates complex pole pairs in the transfer function approximation, at least one complex pole pair is needed to represent low-loss line response. Figure 26-11: shows a typical low-loss line, along with the transfer function sources used to test the various approximations. The transfer functions were obtained by asymptotic waveform evaluation[6].



Figure 26-11: Circuits for a Low-Loss Line

Low-Loss Line Circuit File

```
* Laplace testing LC line Pillage Apr 1990
.Tran 0.02ns 8ns
.Options Post Accurate List Probe
v1 1 0 PWL 0ns 0 0.1ns 0 0.2ns 5
r1 1 2 25
L1 2 3 10nH
c2 3 0 1pF
L2 3 4 10nH
c3 4 0 1pF
```

```
L3 4 5 100nH
c4 5 0 1pF
r4 5 0 400

e3 8 0 LAPLACE 1 0 0.94 / 1.0 0.6n

e2 7 0 LAPLACE 1 0 0.94e20 / 1.0e20 0.348e11 14.8 1.06e-9
2.53e-19
+ SCALE=1.0e-20

e1 6 0 LAPLACE 1 0 0.94 / 1 0.2717e-9 0.12486e-18
.Probe v(1) v(5) v(6) v(7) v(8)
.Print v(1) v(5) v(6) v(7) v(8)
.End
```

Figure 26-12: shows the transient response of the low-loss line, along with E element Laplace models using one, two, and four poles[6]. Note that the single-pole model shows none of the ringing of the higher order models. Also, all of the E models had to adjust the gain of their response for the finite load resistance, so the models are not independent of the load impedance. The 0.94 gain multiplier in the models takes care of the 25 ohm source and 400 ohm load voltage divider. All of the approximations give good delay estimations.

While the two-pole approximation gives reasonable agreement with the transient overshoot, the four-pole model gives almost perfect agreement. The actual circuit has six poles. Scaling was used to bring some of the very small numbers in the Laplace model above the 1e-28 limit of Star-Hspice. The SCALE parameter multiplies every parameter in the LAPLACE specification by the same value, in this case 1.0E-20.



Figure 26-12: – Transient Response of the Low-Loss Line

A low-loss line allows reflections between the load and source, while the loss of an RC line usually isolates the source from the load. So you can either incorporate the load into the AWE transfer function approximation or create an Star-Hspice model that allows source/load interaction. If you allow source/load interaction, the AWE expansions do not have to be done each time you change load impedances, allowing you to handle nonlinear loads and remove the need for a gain multiplier, as in the circuit file shown. You can use four voltage controlled current sources, or G elements, to create a Y-parameter model for a transmission line. The Y-parameter network allows the source/load interaction needed. The next example shows such a Y-parameter model for a low-loss line.

Y-Parameter Line Modeling

A model that is independent of load impedance is more complicated. You can still use AWE techniques, but you need a way for the load voltage and current must be able to interact with the source impedance. Given a transmission line of 100 ohms and 0.4 ns total delay, as shown in Figure 26-13:, compare the response of the line using a Y-parameter model and a single-pole model.



Figure 26-13: Line and Y-Parameter Modeling

The voltage and current definitions for a Y-parameter model are shown in Figure 26-14:.



Figure 26-14: – Y Matrix for the Two-Port Network

The general network in Figure 26-14: is described by the following equations, which can be translated into G elements:

$$I_1 = Y_{11}V_{in} + Y_{12}V_{out}$$

$$I_2 = Y_{21}V_{in} + Y_{22}V_{out}$$

A schematic for a set of two-port Y parameters is shown in Figure 26-15:. Note that the circuit is essentially composed of G elements.



Figure 26-15: – Schematic for the Y-Parameter Network

The Laplace parameters for the Y-parameter model are determined by a Pade expansion of the Y-parameters of a transmission line, as shown in matrix form in the following equation.

$$Y = \frac{1}{Z_o} \cdot \begin{bmatrix} \coth(p) & -\mathrm{csch}(p) \\ -\mathrm{csch}(p) & \coth(p) \end{bmatrix},$$

where p is the product of the propagation constant and the line length[7].

A Pade approximation contains polynomials in both the numerator and the denominator. Since a Pade approximation can model both poles and zeros and since *coth* and *csch* functions also contain both poles and zeros, a Pade approximation gives a better low order model than a series approximation. A Pade expansion of *coth(p)* and *csch(p)*, with second order numerator and third order denominator, is given below:

$$coth(p) \rightarrow \frac{\left(1 + \frac{2}{5} \cdot p^2\right)}{\left(p + \frac{1}{15} \cdot p^3\right)}$$

$$csch(p) \rightarrow \frac{\left(1 - \frac{1}{20} \cdot p^2\right)}{\left(p + \frac{7}{60} \cdot p^3\right)}$$

When you substitute ($s \cdot \mathrm{length} \cdot \sqrt{LC}$) for *p*, you get polynomial expressions for each G element. When you substitute 400 nH for *L*, 40 pF for *C*, 0.1 meter for length, and 100 for Z_o ($Z_o = \sqrt{L/C}$) in the matrix equation above, you get values you can use in a circuit file.

The circuit file shown below uses all of the above substitutions. The Pade approximations have different denominators for *csch* and *coth*, but the circuit file contains identical denominators. Although the actual denominators for *csch* and *coth* are only slightly different, using them would cause oscillations in the Star-Hspice response. To avoid this problem, use the same denominator in the *coth* and *csch* functions in the example. The simulation results may vary, depending on which denominator is used as the common denominator, because the coefficient of the third order term is changed (but by less than a factor of 2).

LC Line Circuit File

```
* Laplace testing LC line Pade
.Tran 0.02ns 5ns
.Options Post Accurate List Probe
v1 1 0 PWL 0ns 0 0.1ns 0 0.2ns 5
r1 1 2 25
r3 1 3 25
u1 2 0 5 0 wire1 L=0.1
r4 5 0 400
r8 8 0 400

e1 6 0 LAPLACE 1 0 1 / 1 0.4n

Gy11 3 0 LAPLACE 3 0 320016 0.0 2.048e-14 / 0.0 0.0128 0.0
2.389e-22
Gy12 3 0 LAPLACE 8 0 -320016 0.0 2.56e-15 / 0.0 0.0128 0.0
2.389e-22
Gy21 8 0 LAPLACE 3 0 -320016 0.0 2.56e-15 / 0.0 0.0128 0.0
2.389e-22
Gy22 8 0 LAPLACE 8 0 320016 0.0 2.048e-14 / 0.0 0.0128 0.0
2.389e-22

.model wire1 U Level=3 PLEV=1 ELEV=3 LLEV=0 MAXL=20
+ ZK=100 DELAY=4.0n

.Probe v(1) v(5) v(6) v(8)
.Print v(1) v(5) v(6) v(8)
.End
```

Figure 26-16: compares the output of the Y-parameter model with that of a full transmission line simulation and with that obtained for a single pole transfer function. In the latter case, the gain was not corrected for the load impedance, so the function produces an incorrect final voltage level. As expected, the Y-parameter model gives the correct final voltage level. Although the Y-parameter model gives a good approximation of the circuit delay, it contains too few poles to model the transient details fully. However, the Y-parameter model does give excellent agreement with the overshoot and settling times.



Figure 26-16: – Transient Response of the Y-Parameter Line Model

Comparison of Circuit and Pole/Zero Models

This example simulates a ninth order low-pass filter circuit and compares the results with its equivalent pole/zero description using an E element. The results are identical, but the pole/zero model runs about 40% faster. The total CPU times for the two methods are shown in “Simulation Time Summary” on page 26-42. For larger circuits, the computation time saving can be much higher.

The input listings for each model type are shown below. Figures 26-17 and 26-18 display the transient and frequency response comparisons resulting from the two modeling methods.

Circuit Model Input Listing

```
low_pass9a.sp 9th order low_pass filter.
* Reference: Jiri Vlach and Kishore Singhal, “Computer
Methods for
* Circuit Analysis and Design”, Van Nostrand Reinhold Co.,
1983,
* pages 142, 494-496.
*
.PARAM freq=100 tstop='2.0/freq'
*.PZ v(out) vin
.AC dec 50 .1k 100k
.OPTIONS dcstep=1e3 post probe unwrap
.PROBE ac vdb(out) vp(out)
.TRAN STEP='tstop/200' STOP=tstop
.PROBE v(out)
vin in GND sin(0,1,freq) ac 1
.SUBCKT fdnr 1 r1=2k c1=12n r4=4.5k
r1 1 2 r1
c1 2 3 c1
r2 3 4 3.3k
r3 4 5 3.3k
r4 5 6 r4
c2 6 0 10n
eop1 5 0 opamp 2 4
eop2 3 0 opamp 6 4
.ENDS
*
rs in 1 5.4779k
r12 1 2 4.44k
r23 2 3 3.2201k
r34 3 4 3.63678k
r45 4 out 1.2201k
c5 out 0 10n
x1 1 fdnr r1=2.0076k  c1=12n    r4=4.5898k
x2 2 fdnr r1=5.9999k  c1=6.8n   r4=4.25725k
x3 3 fdnr r1=5.88327k c1=4.7n   r4=5.62599k
x4 4 fdnr r1=1.0301k  c1=6.8n   r4=5.808498k
.END
```

Pole/Zero Model Input Listing

```
ninth.sp 9th order low_pass filter.
.PARAM twopi=6.2831853072
.PARAM freq=100 tstop='2.0/freq'
.AC dec 50 .1k 100k
.OPTIONS dcstep=1e3 post probe unwrap
.PROBE ac vdb(outp) vp(outp)
.TRAN STEP='tstop/200' STOP=tstop
.PROBE  v(outp)
vin in GND sin(0,1,freq) ac 1
Epole outp GND POLE in GND   417.6153
+    0. 3.8188k
+    0. 4.0352k
+    0. 4.7862k
+    0. 7.8903k / 1.0
+   '73.0669*twopi'     3.5400k
+   '289.3438*twopi'    3.4362k
+   '755.0697*twopi'    3.0945k
+   '1.5793k*twopi'     2.1105k
+   '2.1418k*twopi'     0.
repole outp GND 1e12
.END
```

Simulation Time Summary

Circuit model simulation times:

```
analysis                 time        # points    . iter
conv.iter

op point                 0.23         1          3
ac analysis              0.47         151        151
transient                0.75         201        226          113
rev= 0
readin                   0.22
errchk                   0.13
setup                    0.10
output                   0.00
               total cpu time  1.98 seconds
```

Pole/zero model simulation times:

```
analysis              time        # points     tot. iter
conv.iter

op point              0.12        1            3
ac analysis           0.22        151          151
transient             0.40        201          222          111
rev= 0
readin                0.23
errchk                0.13
setup                 0.02
output                0.00
```

total cpu time 1.23 seconds



Figure 26-17: Transient Responses of the Circuit and Pole/Zero Models



Figure 26-18: AC Analysis Responses of the Circuit and Pole/Zero Models

Modeling Switched Capacitor Filters

This section describes how to create a model

Switched Capacitor Network

It is possible to model a resistor as a capacitor and switch combination. The value of the equivalent is proportional to the frequency of the switch divided by the capacitance.

Construct a filter from MOSFETs and capacitors where the filter characteristics are a function of the switching frequency of the MOSFETs.

In order to quickly determine the filter characteristics, use ideal switches (voltage controlled resistors) instead of MOSFETs. The resulting simulation speedup can be as great as 7 to 10 times faster than a circuit using MOSFETs.

The model constructs an RC network using a resistor and a capacitor along with a switched capacitor equivalent network. Node RCOUT is the resistor/capacitor output, and VCROUT is the switched capacitor output.

The switches GVCR1 and GVCR2, together with the capacitance C3, model the resistor. The resistor value is calculated as:

$$Res = \frac{Tswitch}{C3}$$

where Tswitch is the period of the pulses PHI1 and PHI2.



Figure 26-19: VCR1.SP Switched Capacitor RC Circuit

Example

```
*FILE:VCR1.SP   A SWITCHED CAPACITOR RC CIRCUIT
.OPTIONS acct NOMOD POST

.IC V(SW1)=0 V(RCOUT)=0 V(VCROUT)=0
.TRAN 5U 200U
.GRAPH RC=V(RCOUT)  SWITCH=V(VCROUT)  (0,5)

VCC   VCC     GND     5V

C     RCOUT   GND     1NF
R     VCC     RCOUT   25K

C6    VCROUT  GND     1NF
* equivalent circuit for 25k resistor  r=12.5us/.5nf
VA    PHI1  GND    PULSE 0 5 1US .5US .5US 3US 12.5US
```

```
VB     PHI2   GND     PULSE 0 5 7US .5US .5US 3US 12.5US
GVCR1 VCC SW1 PHI1 GND LEVEL=1 MIN=100 MAX=1MEG 1.MEG -.5MEG
GVCR2 SW1 VCROUT PHI2 GND LEVEL=1 MIN=100 MAX=1MEG 1.MEG .5MEG
C3     SW1     GND     .5NF

.END
```

Switched Capacitor Filter Example - Fifth Order

This example is a fifth order elliptic switched capacitor filter with passband 0-1 kHz, loss less than 0.05 dB. It is realized by cascading linear, high_Q biquad, and low_Q biquad sections. The G element models the switches with a resistance of 1 ohm when the switch is closed and 100 Megohm when it is open. The E element models op-amps as an ideal op-amp. The transient response of the filter is provided for 1 kHz and 2 kHz sinusoidal input signal[8].



Figure 26-20: Linear Section



Figure 26-21: High_Q Biquad Section



Figure 26-22: Low_Q Biquad Section

Star-Hspice Input File for 5th Order Switched Capacitor Filter

```
SWCAP5.SP Fifth Order Elliptic Switched Capacitor Filter.
.OPTIONS POST PROBE
.GLOBAL phi1 phi2
.TRAN 2u 3.2m UIC
*.GRAPH v(phi1) v(phi2) V(in)
.PROBE V(out)
*.PLOT v(in) v(phi1) v(phi2) v(out
*Iin   0    in SIN(0,1ma,1.0khz)
Iin   0    in SIN(0,1v,2khz)
Vphi1 phi1 0  PULSE(0,2 00u,.5u,.5u,7u,20u)
Vphi2 phi2 0  PULSE(0,2 10u,.5u,.5u,7u,20u)
Rsrc  in   0  1k
Rload out  0  1k
Xsh   in   out1 sh
Xlin  out1 out2 linear
Xhq   out2 out3 hqbiq
Xlq   out3 out  lqbiq
```

Sample and Hold

```
.SUBCKT sh in out
Gs1  in  1 VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
Eop1 out 0 OPAMP 1  out
Ch    1  0  1.0pf
.ENDS
```

Linear Section

```
.SUBCKT linear  in out
Gs1  in 1 VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
Gs2  1  0 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
Cs   1  2  1.0pf
Gs3  2  0 VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
Gs4  2  3 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
Eop1 out 0 OPAMP 0 3
Ce   out 3  9.6725pf
Gs5  out 4 VCR PWL(1) phi2 0 0.5v,100meg 1.0v,1.0
Gs6  4   0 VCR PWL(1) phi1 0 0.5v,100meg 1.0v,1.0
Cd   4   2 1.0pf
Csh  in  3 0.5pf
.ENDS
```

High_Q Biquad Section

```
.SUBCKT hqbiq in out
Gs1  in 1 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
Gs2  1  0 VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
C1   1  2  0.5pf
Gs3  2  0 VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
Gs4  2  3 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
Eop1 4  0 OPAMP 0 3
Ca   3  4  7.072pf
Gs5  4  5 VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
Gs6  5  0 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
C3   5  6  0.59075pf
Gs7  6  7 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
Gs8  6  0 VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
Eop2 out 0 OPAMP 0 7
Cb   7  out  4.3733pf
Gs9  out 8 VCR PWL(1) phi2 0 0.5v,100meg 1.0v,1.0
Gs10 8   0 VCR PWL(1) phi1 0 0.5v,100meg 1.0v,1.0
C4   out 3 1.6518pf
C2   8   2 0.9963pf
C11  7  in 0.5pf
.ENDS
```

Low_Q Biquad Section

```
.SUBCKT lqbiq in out
Gs1  in 1 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
Gs2  1  2 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
C1   2  3  0.9963pf
Gs3  2  0 VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
Gs4  3  0 VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
Gs5  3  4 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
Ca   4  5  8.833pf
Eop1 5  0 OPAMP 0 4
Gs6  5  6 VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
Gs7  6  0 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
C3   6  7  1.0558pf
Gs8  7  8 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
Gs9  7  0 VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
Eop2 9  0 OPAMP 0 8
Cb   8  9  3.8643pf
Gs10 9  10 VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
```

```
Gs11 10 0  VCR PWL(1) phi1 0  0.5v,100meg 1.0v,1.0
C4   10 7  0.5pf
C2   10 3  0.5pf
C11  8  1  3.15425pf
Gs12 9 out VCR PWL(1) phi2 0  0.5v,100meg 1.0v,1.0
.ENDS
.END
```



Figure 26-23: Response to 1 kHz Sinusoidal Input



Figure 26-24: Response to 2 kHz Sinusoidal Input

References

References for this chapter are listed below.

1. Williams, Arthur B., and Taylor, Fred J. *Electronic Filter Design Handbook*. New York: McGraw-Hill, 1988, pp. 6-20 to 6-23.
2. Nillson, James W. *Electric Circuits,* 4th Edition. Reading, Massachusetts: Addison-Wesley, 1993.
3. Edminister, Joseph A. *Electric Circuits*. New York: McGraw-Hill, 1965.
4. Ghausi, Kelly, and M.S. *On the Effective Dominant Pole of the Distributed RC Networks*. *Jour. Franklin Inst*., June 1965, pp. 417- 429.
5. Elmore, W.C. and Sands, M. *Electronics, National Nuclear Energy Series*, New York: McGraw-Hill, 1949.
6. Pillage, L.T. and Rohrer, R.A. *Asymptotic Waveform Evaluation for Timing Analysis*, *IEEE Trans. CAD*, Apr. 1990, pp. 352 - 366.
7. Kuo, F. F. *Network Analysis and Synthesis*. John Wiley and Sons, 1966.
8. Gregorian, Roubik & Temes, Gabor C. *Analog MOS Integrated Circuits*. J. Wiley, 1986, page 354.